


Application Note		資料作成 090306	資料番号 an2k-004
テーマ	オムロン 温調器、カウンタ・タイマーとの RS-485 Compoway 通信。		
使用機器	MPC-2000 (MPC-2100), MRS-MCOM 温調器 : オムロン E5EN、カウンタ・タイマ : オムロン H8GN タッチパネル : パナソニック 電工 GT11		

■ 機器構成



■ 概要

- ・ MPC-2000/2100 にはオムロン(株)の Compoway プロトコル通信をマクロ化したコマンドがあります。
- ・ マクロコマンドを使えば文字列の組立てが簡単に、BCC の計算が不要になります。
- ・ MRS-MCOM には RS-485 フェールセーフ回路が内蔵されているので外付回路は不要です(終端 R は付けて下さい)。

■ COMPOWAY マクロコマンドと PRINT#、INPUT# の COMPOWAY オプション

- ・ 送信手順
 - 1 COMPOWAY コマンドで送信するテキストを構築します。
 - 2 PRINT# コマンドに COMPOWAY オプションを与えて実行すると STX と ETX、BCC を付加したコマンドフレームを送信します。
- ・ 受信手順
 - 1 INPUT# コマンドに COMPOWAY オプションを与えて実行するとレスポンスフレームを受信し、BCC を計算します。
 - 2 COMPOWAY コマンドでレスポンスフレームから要素を変数に展開します。

■ 温調器の現在値(温度)読出例

```

/* MRS-MCOM (DSW=6) J6 コネクタ (CH5) を使用します。
CNFG# 5 RS485 "9600b7pes2NONE" /* 通信初期化
/* コマンドフレームのテキスト部分の要素を変数・文字列変数に入れています。
node_no=1 /* ノード no
sub_adr=0 /* サブアドレス
sid=0 /* SID
mrc_src$="0101" /* MRC, SRC
hensu_shu$="C0" /* 変数種別
str_adr$="0000" /* 開始アドレス
bit_ichi$="00" /* bit 位置
yoso_su$="0001" /* 要素数
setteichi$="" /* 設定値 無し
/* コマンドテキスト部分(「サービス要求 PDU」)を作成します
cmd_txt$mrc_src$+hensu_shu$+str_adr$+bit_ichi$+yoso_su$+setteichi$
/* ノード No からコマンドテキストまで結合して snd$ に入れます
COMPOWAY node_no sub_adr sid cmd_txt$ snd$
/* コマンドフレームを送信します
PRINT# 5 COMPOWAY snd$
/* レスポンスフレームを rcv$ に受信します
INPUT# 5 COMPOWAY TMOU|2 rcv$
/* レスポンスフレームから変数・文字列変数に展開します
/* res$ にコマンドテキスト(サービス応答 PDU)の文字列が入ります
COMPOWAY rcv$ node_no sub_adr end_code res$
/* res$ の 0 から数えて 4 文字目から 4 文字がレスポンスコードです
ptr_=res$+4 /* ptr_ はポインタ予約変数。res$ の 4 文字目を指す
res_code=HEX (PTR$(4)) /* ptr_ の位置から 4 文字コピー
/* res$ の 0 から数えて 8 文字目から 8 文字が読出データです
ptr_=res$+8 /* ポインタは res$ の 8 文字目を指す
res_data$=PTR$(8) /* ptr_ の位置から 8 文字コピー
PRINT res_code HEX (res_data$) /* 表示

```

■ サンプルプログラム

- ・ 温調器の本体属性読出、コントローラステータス読出、現在値(温度)読出、カウンタ書込読出を行っています。

```

MEWNET 38400 /* タッチパネル MPC-2000 J1
CNFG# 5 RS485 "9600b7pes2NONE" /* 通信初期化
FORMAT "" /* 文字列フォーマット無し
COUNT_485=0 /* TOTAL COUNT
H8GN_CNT=0 /* カウンタ書込み(max9999)
rs485retry=0 /* リトライカウント
S_MBK 0 200~Wrd /* タッチパネル 温度表示
S_MBK 0 210~Wrd /* タッチパネル TOTAL COUNT
S_MBK 0 220~Wrd /* タッチパネル カウンタ1表示
S_MBK 0 230~Wrd /* タッチパネル カウンタ2表示
S_MBK 0 240~Wrd /* タッチパネル リトライカウント

/* コマンドフレーム共通パラメータ
sub_adr=0 /* サブアドレス
sid=0 /* SID

/* 温調器 本体属性読出、コントローラステータス読出 共通パラメータ
hensu_shu$="" /* 変数種別 無し
str_adr$="" /* 開始アドレス 無し
bit_ichi$="" /* bit位置 無し
yoso_su$="" /* 要素数 無し
setteichi$="" /* 設定値 無し
GOSUB *RS-485_READ_ZOKUSEI 1 /* 本体属性読出 パラメータはノードNo
GOSUB *RS-485_READ_STATUS 1 /* コントローラステータス読出

/* 温調器 カウンタ 変数エリア 書込/読出 コマンドテキスト共通パラメータ
str_adr$="0000" /* 開始アドレス
bit_ichi$="00" /* bit位置
yoso_su$="0001" /* 要素数

DO
COUNT_485=COUNT_485+1 /* TOTAL COUNT
S_MBK COUNT_485 210~Wrd /* タッチパネル表示

GOSUB *RS-485_READ_ONDO 1 /* 温調器 現在温度読出
S_MBK HEX(res_data$) 200 /* 温度表示

H8GN_CNT=H8GN_CNT+1 /* カウンタに書き込む値
IF H8GN_CNT>9999 THEN
H8GN_CNT=0
END_IF

GOSUB *RS-485_WRITE_COUNT 3 /* カウンタ書込
GOSUB *RS-485_READ_COUNT 3 /* カウンタ読出
S_MBK HEX(res_data$) 220 /* cnt1表示

GOSUB *RS-485_WRITE_COUNT 4 /* カウンタ書込
GOSUB *RS-485_READ_COUNT 4 /* カウンタ読出
S_MBK HEX(res_data$) 230 /* cnt2表示
LOOP

*RS-485_READ_ZOKUSEI /* 温調器 本体属性読出
_VAR node_no /* ノード no
mrc_src$="0503" /* MRC, SRC
GOSUB *RS-485_SEND_READ 14 /* 送受信 パラメータは読出データ文字数
PRINT "温調器 形式・バッファサイズ" res_data$
RETURN

*RS-485_READ_STATUS /* 温調器 コントローラステータス読出
_VAR node_no /* ノード no
mrc_src$="0601" /* MRC, SRC
GOSUB *RS-485_SEND_READ 4
PRINT "温調器 運転状態・関連情報" res_data$
RETURN

*RS-485_READ_ONDO /* 温調器 現在温度読出
_VAR node_no /* ノード no
mrc_src$="0101" /* MRC, SRC
hensu_shu$="C0" /* 変数種別
setteichi$="" /* 設定値 無し
GOSUB *RS-485_SEND_READ 8
PRINT node_no "レコード" res_code "温度" HEX(res_data$)
RETURN

*RS-485_WRITE_COUNT /* カウンタ値書込

```

```

_VAR node_no          /* ノード no
mrc_src$="0102"       /* MRC, SRC
hensu_shu$="C2"      /* 変数種別
setteichi$=HEX$(H8GN_CNT) /* 設定値 example H8GN_CNT=1 -> HEX$(H8GN_CNT) -> "00000001"
GOSUB *RS-485_SEND_READ 0
PRINT node_no "レスコード" res_code
RETURN

*RS-485_READ_COUNT   /* カウンタ値読出
_VAR node_no
mrc_src$="0101"      /* MRC, SRC
hensu_shu$="C2"     /* 変数種別
setteichi$=""       /* 設定値 無し
GOSUB *RS-485_SEND_READ 8
PRINT node_no "レスコード" res_code "カウンタ" HEX(res_data$)
RETURN

*RS-485_SEND_READ    /* データ構築→送信→受信
_VAR data_len        /* 読出データ長
cmd_txt$=mrc_src$+hensu_shu$+str_adr$+bit_ichi$+yoso_su$+setteichi$ /* コマンドテキスト
COMPOWAY node_no sub_adr sid cmd_txt$ snd$ /* ノード No からコマンドテキストまで結合
*RS-485_SEND_READ_RETRY
S_MBK rs485retry 240 /* タッチパネル表示
PRINT# 5 COMPOWAY snd$ /* コマンドフレーム構築 (STX, ETX, BCC 付加)→送信
INPUT# 5 COMPOWAY TMOU|2 rcv$ /* レスポンスフレーム受信 (BCC チェック)
PRINT "snd$rcv$" snd$ rcv$
IF rse_<>0 THEN /* rse_は予約変数。TMOU の場合 rse_=4 になる
PRINT "communication error" rse_
INPUT# 5 CLR_BUF
rs485retry=rs485retry+1 /* Retry Count
GOTO *RS-485_SEND_READ_RETRY
END_IF
COMPOWAY rcv$ node_no sub_adr end_code res$ /* レスポンスフレームから展開
ptr_=$+4 /* res$にレスポンスコードと読出データ
res_code=HEX (PTR$(4)) /* res$ の中の'レスポンスコード' 4文字
ptr_=$+8
res_data$=PTR$(data_len) /* res$ の中の'読出データ' data_len 文字
RETURN

```

■その他

- ・ Compoway 詳細は機器の通信マニュアル、ユーザズマニュアルをご覧ください。

以上